
Fonctions arithmétiques
La division euclidienne s'obtient avec les commandes //  (donne le quotient) et %  (donne le reste).

In [1]: 9//4

In [2]: 9%4

Ces commandes marchent aussi avec des polynômes !

In [3]: R.<X> = PolynomialRing(QQ)
P = X^3+3*X+1
Q = X+4
P//Q

In [4]: P%Q

In [5]: (X^2-4*X+19)*(X+4) - 75

Les ensembles usuels sont obtenus avec les commandes NN  (entiers naturels), ZZ  (entiers relatifs), QQ  (nombres rationels), RR  (nombres réels), CC
(nombres complexes). En particulier, la commande suivante force le nombre  à être considéré comme un réel :3

In [6]: RR(3)

L'anneau des entiers modulo n est obtenu par la commande Zmod(n) .

In [7]: Z13 = Zmod(13)
Z13(17)

Si on veut faire une exponentiation dans l'anneau , on peut utiliser la fonction power_mod .ℤ/nℤ

In [8]: power_mod(4,4,13)

In [9]: 4^4 % 13

Le pgcd de deux entiers est obtenu avec la commande gcd . Cela marche aussi avec des polynômes (c'est vrai pour de nombreuses commandes dans ce qui
suit).

In [10]: gcd(84,120)

La commande xgcd(a,b)  renvoie le pgcd de  et , ainsi que deux nombres  et  tels que .a b u v au + bv = pgcd(a, b)

In [11]: xgcd(84,120)

In [12]: 3*84-2*120

Pour résoudre un système chinois, on utilise la commande crt(L1,L2) , où L1  et L2  sont deux listes de même taille donnant les équations de
congruence. Par exemple, crt([3,2],[7,11])  résoud le système d'équations  et .x ≡ 3 mod 7 x ≡ 2 mod 11

In [13]: crt([3,2],[7,11])

Pour savoir si un nombre  est un carré dans , on utilise le symbole de Legendre , égal à  si  divise , à  si  est un carré non nul modulo , et à 
 sinon. La commande Sage est legendre_symbol(k,p) .

k ℤ/pℤ ( )k
p 0 p k 1 k p

In [14]: legendre_symbol(10,13)

In [15]: (6^2)%13

La fonction is_prime(a)  teste si un nombre  est premier. La fonction next_prime(a)  renvoie le plus petit nombre premier suivant , la fonction 
random_prime(a)  renvoie un nombre premier aléatoire plus petit que . Finalement, prime_range(a,b)  donne la liste de tous les nombres premiers

entre  et .

a a
a

In [16]: is_prime(113)

In [17]: next_prime(2^16)

In [18]: random_prime(2^16)

In [19]: prime_range(50,100)

Pour décomposer un nombre en facteurs premiers, on utilise factor  (marche aussi pour trouver les facteurs irréductibles d'un polynôme).

In [20]: factor(2^32+1)

In [21]: R.<X> = PolynomialRing(QQ)
P = X^3-3*X+2
factor(P)

Anneaux de polynômes, quotients, corps finis
Pour définir un anneau de polynômes , on doit indiquer à Sage le corps k  et le nom de la variable X . Ceci est fait simultanément avec la commande
suivante : Rk.<X>=PolynomialRing(k) . Ceci crée une variable appelée X , ainsi que l'objet Rk  qui est l'anneau des polynômes en une variable sur le
corps . On peut bien sûr utiliser une autre nom pour la variable, choisir le corps de base  que l'on souhaite (ou même utiliser un anneau  et travailler avec
un anneau de polynômes ) ; il est également possible de travailler avec des polynômes en plusieurs variables.

k[X]

k k A

In [22]: RQ.<X> = PolynomialRing(QQ)
RQ

In [23]: X^2+3*X-2/3

In [24]: _.parent()

In [25]: RQ2.<X,Y> = PolynomialRing(QQ,2)

In [26]: RQ2

In [27]: X*Y+3*Y^3

In [28]: _.parent()

On travaillera quasiment exclusivement avec des polynômes en une variable. Dans ce cas, une convention utile est d'appeler Rk  l'anneau des polynômes à
coefficients dans k . Pour savoir si un polynôme  est irréductible dans , on utilise la méthode is_irreducible . Attention, la réponse dépend du
corps .

P k[X]
k

In [29]: RR.<X> = PolynomialRing(RR)
P = X^2+1
P.is_irreducible()

In [30]: RC.<X> = PolynomialRing(CC)
P = X^2+1
P.is_irreducible()

In [31]: factor(P)

Étant donné un anneau de polynômes  et un polynôme , il est possible de définir dans Sage l'anneau quotient , en utilisant la méthode 
quotient  des anneaux de polynômes. La manipulation de ces anneaux est similaire à celle des anneaux .

k[X] P(X) k[X]/(P)
ℤ/nℤ

In [32]: RQ.<X> = PolynomialRing(QQ)
P(X)=X^2+3*X-2
RQquotP = RQ.quotient(P)

In [33]: RQquotP

In [34]: RQquotP(X^3-1)

Par convention, si X  est le nom de la variable dans , alors Sage appelle Xbar  l'image de cette variable dans le quotient .k[X] k[X]/(P)

In [35]: RQ(X^3-1) % RQ(X^2+3*X-2)

Ici, le polynôme  est irréductible dans , donc l'anneau quotient est un corps.P ℚ[X]

In [36]: RQquotP.is_field()

Le corps fini  avec  peut être construit en prenant le quotient de  par l'idéal engendré par un polynôme irréductible  de degré . La
commande GF(q)  définit ce corps dans Sage ; si l'on veut préciser le nom u  de l'image de  par le morphisme canonique , on
écrira : Fq.<u> = GF(q) . Ceci crée le corps fini (nommé Fq ) par Sage, et indique le nom u  du générateur de  sur .

𝔽q q = pn [X]𝔽p P n
X [X] → [X]/(P) =𝔽p 𝔽p 𝔽q

𝔽q 𝔽p

In [37]: F16.<u> = GF(16)
F16

Pour savoir quel polynôme irréductible Sage a choisi, on utilise la méthode polynomial  :

In [38]: F16.polynomial()

Si au moment de la création du corps fini on veut préciser quel polynôme irréductible utiliser, on peut le faire avec l'option modulus  (le polynôme indiqué doit
alors utiliser la variable x ).

In [39]: F16_2.<v> = GF(16, modulus=x^4+x^3+1)
F16_2, F16_2.polynomial()

Le morphisme de Frobenius de  avec  est l'application  ; c'est un automorphisme de corps. On l'obtient dans Sage à partir de  avec la
méthode frobenius_endomorphism .

𝔽q q = pn≥1 x ↦ xp 𝔽q

In [40]: F = F16.frobenius_endomorphism()
F

In [42]: F(u^3+u+1)

In [43]: F(u^3+u+1),F(F(u^3+u+1)),F(F(F(u^3+u+1))),F(F(F(F(u^3+u+1))))

Étant donné un polynôme  à coefficients dans , le morphisme de Frobenius permute les racines de  dans le corps . Considérons par exemple le
polynôme  dans . Ce polynôme est irréductible dans .

P 𝔽p P 𝔽q
P(X) = + + 1X4 X3 [X]𝔽2 𝔽2

In [45]: RF2.<X> = PolynomialRing(GF(2))
P = RF2(X^4+X^3+1)
P

In [46]: P.is_irreducible()

Par contre,  n'est pas irréductible dans  ; il y est même scindé à racines simples, le morphisme de Frobenius permutant cycliquement les racines.P 𝔽16

In [47]: RF16.<Y> = PolynomialRing(F16)
PP = RF16(P)
PP

In [48]: factor(PP)

Résultant de deux polynômes
Le résultant de deux polynômes en une variable permet de savoir s'ils ont un facteur irréductible en commun (il s'annule si et seulement si c'est le cas). La
commande Sage est P.resultant(Q) .

In [56]: RQ.<X> = PolynomialRing(QQ)
P = X^4 + 3*X^3 - 5*X - 3
Q = X^5 + X^4 + 2*X^2 - 10*X + 3
P.resultant(Q)

In [57]: gcd(P,Q)

Une application intéressante de la théorie des résultants est la détermination des intersections de deux courbes algébriques. Considérons les deux courbes
algébriques réelles  et  déterminées par les équations

et

On peut dessiner ces courbes algébriques avec Sage.

P Q
P(x, y) = − + − 2 y + = 0y4 y3 y2 x2 x4

Q(x, y) = y − 2 = 0.x2

In [76]: RQ2.<x,y> = PolynomialRing(QQ,2)
G = Graphics()
P = RQ2(y^4 - y^3 + y^2 - 2*(x^2)*y + x^4)
Q = RQ2(y - 2*x^2)
G += implicit_plot(P, (x,-1.5,1.5), (y,-0.5,2))
G += implicit_plot(Q, (x,-1.5,1.5), (y,-0.5,2), color="red")
G += points([(0,0),(0.5,0.5),(-0.5,0.5)], size=30, color="black", zorder=10)
G.show(axes=True)

Il semble qu'il y ait trois points d'intersection ; expliquons comment démontrer cela et comment trouver les intersections. Si  est l'ordonnée d'un point
d'intersection  de  et , alors les deux polynômes en une variable  et  ont pour racine commune , donc

L'équation ci-dessus est polynomiale en , et permet de trouver les ordonnées possibles . Pour notre exemple, on procède comme suit :

y0
( , )x0 y0 P Q = P(⋅, )Py0

y0 = Q(⋅, )Qy0
y0 y0

Res( , ) = 0.Py0
Qy0

y0 y0

In [78]: R = P.resultant(Q,x)
R, factor(R)

La commande P.resultant(Q,x)  calcule le résultant par rapport à la variable  des deux polynômes ; on obtient un polynôme en  dont les racines sont
les ordonnées  possibles pour un point d'intersection (on pourrait aussi d'abord chercher les  possibles avec la commande P.resultant(Q,y) ). Ici,
les  possibles sont  et .

x y
y0 x0

y0 = 0y0 =y0
1
2

In [83]: RQX.<x> = PolynomialRing(QQ)
RQY.<y> = PolynomialRing(QQ)
RQY(R).roots()

Les multiplicités  que l'on obtient ont un sens géométrique : le point d'intersection de  et  d'ordonnée  est un point où les courbes sont tangentes et
coincident jusqu'à l'ordre , tandis que les deux points d'intersection d'ordonnée  sont des points de tangence des deux courbes.

Pour trouver les  correspondants aux  précédemment calculées, on peut utiliser la méthode .subs(y=a)  pour obtenir des polynômes en une seule
variable  (élimination de la variable ), et chercher leurs racines.

4 P Q 0
4 1

2

x0 y0
x y

In [84]: Q0 = RQX(Q.subs(y=0))
Q0, Q0.roots()

Pour , on obtient .= 0y0 = 0x0

In [85]: Qhalf = RQX(Q.subs(y=1/2))
Qhalf, Qhalf.roots()

Pour , on obtient . Conclusion : les trois points d'intersection sont ,  et .=y0
1
2 = ±x0

1

In [ ]:  

Out[1]: 2

Out[2]: 1

Out[3]: X^2 - 4*X + 19

Out[4]: -75

Out[5]: X^3 + 3*X + 1

Out[6]: 3.00000000000000

Out[7]: 4

Out[8]: 9

Out[9]: 9

Out[10]: 12

Out[11]: (12, 3, -2)

Out[12]: 12

Out[13]: 24

Out[14]: 1

Out[15]: 10

Out[16]: True

Out[17]: 65537

Out[18]: 32587

Out[19]: [53, 59, 61, 67, 71, 73, 79, 83, 89, 97]

Out[20]: 641 * 6700417

Out[21]: (X + 2) * (X - 1)^2

Out[22]: Univariate Polynomial Ring in X over Rational Field

Out[23]: X^2 + 3*X - 2/3

Out[24]: Univariate Polynomial Ring in X over Rational Field

Out[26]: Multivariate Polynomial Ring in X, Y over Rational Field

Out[27]: 3*Y^3 + X*Y

Out[28]: Multivariate Polynomial Ring in X, Y over Rational Field

Out[29]: True

Out[30]: False

Out[31]: (X - I) * (X + I)

Out[33]: Univariate Quotient Polynomial Ring in Xbar over Rational Field with modulus X^2 + 3*X - 2

Out[34]: 11*Xbar - 7

Out[35]: 11*X - 7

Out[36]: True

Out[37]: Finite Field in u of size 2^4

Out[38]: u^4 + u + 1

Out[39]: (Finite Field in v of size 2^4, v^4 + v^3 + 1)

Out[40]: Frobenius endomorphism u |--> u^2 on Finite Field in u of size 2^4

Out[42]: u^3 + 1

Out[43]: (u^3 + 1, u^3 + u^2 + 1, u^3 + u^2 + u, u^3 + u + 1)

Out[45]: X^4 + X^3 + 1

Out[46]: True

Out[47]: Y^4 + Y^3 + 1

Out[48]: (Y + u^3 + 1) * (Y + u^3 + u + 1) * (Y + u^3 + u^2 + 1) * (Y + u^3 + u^2 + u)

Out[56]: 0

Out[57]: X^2 + X - 3

Out[78]: (16*y^8 - 32*y^7 + 24*y^6 - 8*y^5 + y^4, y^4 * (2*y - 1)^4)

Out[83]: [(1/2, 4), (0, 4)]

Out[84]: (-2*x^2, [(0, 2)])

Out[85]: (-2*x^2 + 1/2, [(1/2, 1), (-1/2, 1)])


